NAG Library Routine Document

C05BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05BAF returns the real values of Lambert's W function $W(x)$, via the routine name.

2 Specification

double precision `FUNCTION C05BAF(X, BRANCH, OFFSET, IFAIL)`

```
INTEGER                  BRANCH, IFAIL
double precision         X
LOGICAL                  OFFSET
```

3 Description

C05BAF calculates an approximate value for the real branches of Lambert's W function (sometimes known as the 'product log' or 'Omega' function), which is the inverse function of

$$f(w) = we^w \qquad \text{for} \qquad w \in C.$$

The function f is many-to-one, and so, except at 0, W is multivalued. C05BAF restricts W and its argument x to be real, resulting in a function defined for $x \geq -\exp(-1)$ and which is double valued on the interval $(-\exp(-1), 0)$. This double-valued function is split into two real-valued branches according to the sign of $W(x) + 1$. We denote by W_0 the branch satisfying $W_0(x) \geq -1$ for all real x, and by W_{-1} the branch satisfying $W_{-1}(x) \leq -1$ for all real x. You may select your branch of interest using the parameter BRANCH.

The precise method used to approximate W is described fully in Barry *et al.* (1995). For x close to $-\exp(-1)$ greater accuracy comes from evaluating $W(-\exp(-1) + \Delta x)$ rather than $W(x)$: by setting OFFSET = .TRUE. on entry you inform C05BAF that you are providing Δx, not x, in X.

4 References

Barry D J, Culligan–Hensley P J, and Barry S J (1995) Real Values of the W-function *ACM Trans. Math. Software* **21 (2)** 161–171

5 Parameters

1: X – ***double precision*** *Input*

On entry: if OFFSET = .TRUE., X is the offset Δx from $-\exp(-1)$ of the intended argument to W; that is, $W(\beta)$ is computed, where $\beta = -\exp(-1) + \Delta x$.

If OFFSET = .FALSE., X is the argument x of the function; that is, $W(\beta)$ is computed, where $\beta = x$.

Constraints:

if BRANCH $= 0$, $-\exp(-1) \leq \beta$;
if BRANCH $= -1$, $-\exp(-1) \leq \beta < 0$.

2: BRANCH – INTEGER *Input*

On entry: the real branch required.

BRANCH $= 0$
The branch W_0 is selected.

BRANCH $= -1$
The branch W_{-1} is selected.

Constraint: BRANCH $= 0$ or -1.

3: OFFSET – LOGICAL *Input*

On entry: controls whether or not X is being specified as an offset from $-\exp(-1)$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: C05BAF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL $= 1$

An input parameter is invalid. If IFAIL $= 0$ or -1 on entry, the output message provides more details of the nature of the warning.

IFAIL $= 2$

Warning: the actual argument to W was very close to $-\exp(-1)$. The output message provides more details of the nature of the warning.

7 Accuracy

For a high percentage of legal X on input, C05BAF is accurate to the number of decimal digits of precision on the host machine (see X02BEF). An extra digit may be lost on some implementations and for a small proportion of such X. This depends on the accuracy of the base-10 logarithm on your system.

8 Further Comments

None.

9 Example

This example reads from a file the values of the required branch, whether or not the arguments to W are to be considered as offsets to $-\exp(-1)$, and the arguments X themselves. It then evaluates the function for these sets of input data X and prints the results.

9.1 Program Text

```
*     C05BAF Example Program Text
*     Mark 22 Release. NAG Copyright 2008.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION W, X
      INTEGER          BRANCH, IFAIL
      LOGICAL          OFFSET
*     .. External Functions ..
      DOUBLE PRECISION C05BAF
      EXTERNAL         C05BAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C05BAF Example Program Results'
*
*     Skip heading in data file
*
      READ (NIN,*)
      READ (NIN,*) BRANCH
      READ (NIN,*) OFFSET
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'BRANCH = ', BRANCH
*
      IF (OFFSET) THEN
         WRITE (NOUT,99997) 'OFFSET = .TRUE.'
      ELSE
         WRITE (NOUT,99997) 'OFFSET = .FALSE.'
      END IF
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) '        X           W(X)     IFAIL'
      WRITE (NOUT,*)
*
   20 CONTINUE
*
      READ (NIN,*,END=40) X
*
      IFAIL = 1
*
      W = C05BAF(X,BRANCH,OFFSET,IFAIL)
*
      IF (IFAIL.GE.0) THEN
         WRITE (NOUT,99999) X, W, IFAIL
         GO TO 20
      ELSE
         WRITE (NOUT,99996) IFAIL
      END IF
*
   40 CONTINUE
*
99999 FORMAT (1X,1P,2(1X,E13.5),1X,I3)
99998 FORMAT (1X,A,I3)
99997 FORMAT (1X,A)
99996 FORMAT (1X,'** C05BAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
C05BAF Example Program Data
  0                                                    : BRANCH
  .FALSE.                                              : OFFSET
  0.5
  1.0
  4.5
  6.0
  7.0D7                                                : X
```

9.3 Program Results

```
 C05BAF Example Program Results

 BRANCH =   0
 OFFSET = .FALSE.

         X           W(X)     IFAIL

    5.00000E-01   3.51734E-01   0
    1.00000E+00   5.67143E-01   0
    4.50000E+00   1.26724E+00   0
    6.00000E+00   1.43240E+00   0
    7.00000E+07   1.53339E+01   0
```